
An analogy

Natural numbers (aka Peano numerals, unary numbers etc):

data Nat = Zero | Succ Nat

plus :: Nat → Nat → Nat
plus Zero n2 = n2
plus (Succ n1) n2 = Succ (plus n1 n2)

Lists (aka stacks, sequences etc):

data List α = Nil | Cons α (List α)

append :: ∀α .List α → List α → List α
append Nil x2 = x2
append (Cons a x1) x2 = Cons a (append x1 x2)

☞ There is a strong analogy between representations of numbers (n) and
representations of container objects (of size n).

Numerical representations

☞ Data structures that are designed on the basis of this analogy are called
numerical representations.

Idea: the data structure inherits the properties of the number system. The
operations on the data structure are modelled after their numerical counterparts.

increment n + 1 insertion into a container
decrement n − 1 deletion from a container
addition n1 + n2 union or merge of two container objects

☞ This design technique is suitable for implementing arbitrary abstractions:
sequences, priority queues, sets etc.

☞ The numerical representations we shall introduce are fully persistent: updates
never destroy the original data structure.

Numerical representations—continued

arithmetic shift n ∗ b we shall see
n / b

multiplication n ∗ k
division n / k split of a container object
number conversion construction of a container object

conversion between different container types

History

I Clancy, Knuth, A programming and problem-solving seminar, 1977.

I Guibas, McCreight, Plass, Roberts, A new representation for linear lists, 1977.

I Vuillemin, A data structure for manipulating priority queues, 1978.

I Okasaki, Purely Functional Data Structures, 1998.

☞ Some material is taken from Okasaki’s book, which I highly recommend.

Outline of the talk

✖ Exploring the analogy (7–27)

✖ A toolbox of number systems (29–41)

✖ Analysis of data structures (43–51)

✖ A worked-out example: 2-3 finger trees (53–64)

Random-access lists

Lists are based on the unary number system; random-access lists are based on the
binary number system.

data Seq α = Nil
| Zero ( Seq (α, α))
| One (α, Seq (α, α))

☞ The type of elements changes from position to position: the top-level
possibly contains an element of type α, the next of type (α, α), the next of type
((α, α), (α, α)) and so on. In other words, n ∗ 2 corresponds to pairing.

☞ Seq is an example of a non-regular or nested data type.

Random-access lists—examples

Nil
One (11,Nil)
Zero (One ((10, 11),Nil))
One (9,One ((10, 11),Nil))
Zero (Zero (One (((8, 9), (10, 11)),Nil)))
One (7,Zero (One (((8, 9), (10, 11)),Nil)))
Zero (One ((6, 7),One (((8, 9), (10, 11)),Nil)))
One (5,One ((6, 7),One (((8, 9), (10, 11)),Nil)))
Zero (Zero (Zero (One ((((4, 5), (6, 7)), ((8, 9), (10, 11))),Nil))))
One (3,Zero (Zero (One ((((4, 5), (6, 7)), ((8, 9), (10, 11))),Nil))))
Zero (One ((2, 3),Zero (One ((((4, 5), (6, 7)), ((8, 9), (10, 11))),Nil))))
One (1,One ((2, 3),Zero (One ((((4, 5), (6, 7)), ((8, 9), (10, 11))),Nil))))

Random-access lists—insertion

Insertion corresponds to binary increment, except that the carry is explicit—the
carry is witnessed by a container object of the appropriate size.

cons :: ∀α . (α, Seq α) → Seq α
cons (a,Nil) = One (a,Nil)
cons (a,Zero x ) = One (a, x )
cons (a1,One (a2, x )) = Zero (cons ((a1, a2), x ))

☞ cons requires a non-schematic form of recursion, called polymorphic
recursion: the recursive call inserts a pair not an element.

☞ cons runs in Θ(log n) worst-case time.

Random-access lists—deletion

Deletion corresponds to binary decrement, except that the borrow is explicit.

uncons :: ∀α . Seq α → (α, Seq α)
uncons (One (a,Nil)) = (a,Nil)
uncons (One (a, x )) = (a,Zero x )
uncons (Zero x ) = let ((a1, a2), x ) = uncons x in (a1,One (a2, x ))

☞ uncons is the mirror image of cons .

cons :: ∀α . (α, Seq α) → Seq α
cons (a,Nil) = One (a,Nil)
cons (a,Zero x ) = One (a, x )
cons (a1,One (a2, x )) = Zero (cons ((a1, a2), x ))

Random-access lists—indexing

Indexing corresponds to . . . (well, it’s a bit like ‘6’).

lookup :: ∀α . Integer → Seq α → α
lookup 0 (One (a, x )) = a
lookup (n + 1) (One (a, x )) = lookup n (Zero x )
lookup (2 ∗ n + 0) (Zero x ) = fst (lookup n x )
lookup (2 ∗ n + 1) (Zero x ) = snd (lookup n x )

Random-access lists—construction

Container objects can be constructed in at least two different ways:

I construct a container object containing n copies of a given element:

replicate :: ∀α . Integer → α → Seq α

I construct a container object from a given list of elements:

toSeq :: ∀α . [α ] → Seq α

Often, the former operation can be implemented more efficiently.

☞ In both cases, construction corresponds to conversion of number
representations: here from the unary to the binary number system.

Conversion of number representations

There are at least two ways to convert a number in one system to the equivalent
number in another system:

I use the arithmetic of the target number system; this is sometimes called the
expansion method; functions of this type are typically folds.

I use the arithmetic of the source number system; this is sometimes called the
multiplication or division method; functions of this type are typically unfolds.

Construction—replicate

Using the arithmetic of the target system (unary to binary):

replicate :: ∀α . Integer → α → Seq α
replicate 0 a = Nil
replicate (n + 1) a = cons (a, replicate n a)

☞ replicate runs in Θ(n) worst-case time; it is not polymorphically recursive.

Construction—replicate—continued

Using the arithmetic of the source system (unary to binary):

replicate :: ∀α . Integer → α → Seq α
replicate n a = if n 0 then Nil

else case modDiv n 2 of
(0, q) → Zero ( replicate q (a, a))
(1, q) → One (a, replicate q (a, a))

☞ replicate runs in Θ(log n) worst-case time; it is polymorphically recursive.

Construction—toSeq

Using the arithmetic of the target system (unary to binary):

toSeq :: ∀α . [α ] → Seq α
toSeq [ ] = Nil
toSeq (a : x ) = cons (a, toSeq x )

☞ toSeq runs in Θ(n) worst-case time.

☞ [α ] is the built-in list data type, which is isomorphic to List α (see page 2):
[α ] ∼= List α, [ ] ∼= Nil , and a : x ∼= Cons a x .



17 JJ J I II 2

Random-access lists—conversion—continued

Using the arithmetic of the source system (unary to binary):

data Digit α = Zero ′ | One ′ α

modDiv2 :: [α ] → (Digit α, [(α, α)])
modDiv2 [ ] = (Zero ′, [ ])
modDiv2 (a : x ) = case modDiv2 x of

(Zero ′, q) → (One ′ a, q)
(One ′ a ′, q) → (Zero ′, (a, a ′) : q)

toSeq :: ∀α . [α ] → Seq α
toSeq x = if null x then Nil

else case modDiv2 x of
(Zero ′, q) → Zero ( toSeq q)
(One ′ a, q) → One (a, toSeq q)

☞ toSeq runs in Θ(n) worst-case time.

Exercises

Exercise 1. Implement two versions of

size :: ∀α . Seq α → Integer

fromSeq :: ∀α . Seq α → [α ]

and determine the worst-case running times (binary to unary).

1-2 random-access lists

The container object that corresponds to ‘0’ contains no elements. This is
wasteful!

☞ Interestingly, we can also use the digits 1 and 2 instead of 0 and 1 (the base
is still 2).

data Seq α = Nil
| One (α, Seq (α, α))
| Two ((α, α), Seq (α, α))

☞ Each number has a unique representation in this system; this is a so-called
zeroless number system.

1-2 random-access lists—examples

Nil
One (11,Nil)
Two ((10, 11),Nil)
One (9,One ((10, 11),Nil))
Two ((8, 9),One ((10, 11),Nil))
One (7,Two (((8, 9), (10, 11)),Nil))
Two ((6, 7),Two (((8, 9), (10, 11)),Nil))
One (5,One ((6, 7),One (((8, 9), (10, 11)),Nil)))
Two ((4, 5),One ((6, 7),One (((8, 9), (10, 11)),Nil)))
One (3,Two (((4, 5), (6, 7)),One (((8, 9), (10, 11)),Nil)))
Two ((2, 3),Two (((4, 5), (6, 7)),One (((8, 9), (10, 11)),Nil)))
One (1,One ((2, 3),Two ((((4, 5), (6, 7)), ((8, 9), (10, 11))),Nil)))

1-2 random-access lists—insertion

cons :: ∀α . (α, Seq α) → Seq α
cons (a,Nil) = One (a,Nil)
cons (a1,One (a2, x )) = Two ((a1, a2), x )
cons (a1,Two ((a2, a3), x )) = One (a1, cons ((a2, a3), x ))

1-2 random-access lists—deletion

Again, uncons is the mirror image of cons .

uncons :: ∀α . Seq α → (α, Seq α)
uncons (One (a,Nil)) = (a,Nil)
uncons (Two ((a1, a2), x )) = (a1,One (a2, x ))
uncons (One (a1, x )) = (a1,Two (uncons x ))

☞ The term Two (uncons x ) corresponds to an arithmetic shift (n ∗ 2):

zero :: ∀α . Seq (α, α) → Seq α
zero x = Two (uncons x )

1-2 random-access lists—conversion

Using zero we can easily convert a 0-1 random-access list into a 1-2
random-access list.

toSeq :: ∀α . Seq01 α → Seq α
toSeq Nil 01 = Nil
toSeq (Zero01 x ) = zero (toSeq x )
toSeq (One01 (a, x )) = One (a, toSeq x )

☞ This implementation uses the arithmetic of the target number system.

Exercises

Exercise 2. Implement a version of toSeq that uses the arithmetic of the source
number system.

Exercise 3. Re-implement indexing for 1-2 random-access lists and show that it
has a running time of Θ(log i) rather than Θ(log n).

0-1-2 random-access lists

If considered in isolation, insertion and deletion both have an amortised running
time of Θ(1). If ‘cons ’ and ‘uncons ’ are mixed, however, the running time
degrades to Θ(log n). Consider:

decr (incr (222222))

This can be remedied using a redundant number system, which employs, for
instance, the digits 0, 1, and 2 (or alternatively the digits 1, 2, and 3):

data Seq α = Nil
| Zero ( Seq (α, α))
| One (α, Seq (α, α))
| Two ((α, α), Seq (α, α))

0-1-2 random-access lists—insertion

We classify 0 and 2 as dangerous and 1 as safe; ‘cons ’ and ‘uncons ’ recurse on
dangerous digits, but always leave a safe digit behind, so that the next operation
to reach that digit will not propagate.

cons :: ∀α . (α, Seq α) → Seq α
cons (a,Nil) = One (a,Nil)
cons (a,Zero x ) = One (a, x )
cons (a1,One (a2, x )) = Two ((a1, a2), x )
cons (a1,Two ((a2, a3), x )) = One (a1, cons ((a2, a3), x ))

0-1-2 random-access lists—deletion

uncons :: ∀α . Seq α → (α, Seq α)
uncons (One (a,Nil)) = (a,Nil)
uncons (One (a, x )) = (a,Zero x )
uncons (Two ((a1, a2), x )) = (a1,One (a2, x ))
uncons (Zero x )

= let ((a1, a2), x ) = uncons x in (a1,One (a2, x ))

☞ uncons is no longer the mirror image of cons .

Outline of the talk

✔ Exploring the analogy (7–27)

✖ A toolbox of number systems (29–41)

✖ Analysis of data structures (43–51)

✖ A worked-out example: 2-3 finger trees (53–64)

Positional number systems

The most common number systems are positional number systems.

(d0 . . . dn−1) =

n−1∑
i=0

di · wi with di ∈ Di.

Unary number system: wi = 1 and Di = {1} for all i.

Binary number system: wi = 2i and Di = {0, 1} for all i.

1-2 binary system: wi = 2i and Di = {1, 2} for all i.

0-1-2 binary system: wi = 2i and Di = {0, 1, 2} for all i.

Common building blocks

complete leaf tree complete tree

pennant binomial tree

Common abstractions

The most elementary operation is adding two 1s of weight wi to form a 1 of
weight wi+1.

I Sequences: if we use leaf trees, adding two 1s amounts to pairing (adding a
new root node), see above.

I Priority queues (bags): we can either use pennants (semi-heaps) or binomial
trees (heaps). For the former, adding two 1s is given by

t uu t ut

a b
a

b a

b

]a6b⇐= a>b=⇒

I Sets: adding two 1s is difficult (if not impossible), since the keys in a search
tree must be maintained in symmetric order, but see below.

Ternary number systems

Number systems with a larger base are often advantageous, because fewer digits
are needed to represent each number. On the downside, processing a single digit
may take longer.

data Seq α = Nil
| Zero ( Seq (α, α, α))
| One (α, Seq (α, α, α))
| Two (α, α, Seq (α, α, α))

☞ We use triples instead of pairs.

Ternary number systems—insertion

cons :: ∀α . (α, Seq α) → Seq α
cons (a,Nil) = One (a,Nil)
cons (a,Zero x ) = One (a, x )
cons (a1,One (a2, x )) = Two (a1, a2, x )
cons (a1,Two (a2, a3, x )) = Zero (cons ((a1, a2, a3), x ))

Fibonacci number system

Fibonacci number system: wi = Fi and Di = {0, 1} for all i, where

F0 = 1
F1 = 2
Fn+2 = Fn + Fn+1

☞ This number system is redundant: we have, for instance, (11) = (001).
However, if we disallow consecutive 1s, then we regain uniqueness.

The fibonacci number system is a lot like the binary number system, except that
it involves only addition and subtraction, not multiplication or division by 2.

Fibonacci number system—continued

The following data type captures the invariant that the digit 1 is not followed by a
second 1.

type Seq α = Fib α (α, α)

data Fib α β = Nil
| Zero ( Fib β (α, β))
| One (α,Fib (α, β) (β, (α, β)))

The smart constructor one restores the invariant:

one :: ∀α β . (α,Fib β (α, β)) → Fib α β
one (a,Nil) = One (a,Nil)
one (a,Zero x ) = One (a, x )
one (a,One (b, x )) = Zero (Zero (one ((a, b), x )))

Fibonacci number system—insertion

Insertion is easy to define using one:

cons :: ∀α . α → Seq α → Seq α
cons a Nil = One (a,Nil)
cons a (Zero x ) = one (a, x )
cons a (One (b, x )) = Zero (one ((a, b), x ))

☞ The type Fib α β is a bit inconvenient to work with because it has two type
arguments.

Factorial number system

Factorial number system: wi = i! and Di = {0, . . . , i} for all i.

☞ Note that
∑n−1

i=0 i · i! = n!− 1.

Application: Given a sequence of elements, generate a random permutation.

I generate a random number r with 0 6 r 6 n !−1;

I convert r into the factorial number system: (d0 . . . dn−1);

I for i = 0 to n− 1: exchange elements at positions i and i + dn−1−i.

0 1 2 3
0 ↔ 0 or 1 or 2 or 3
1 ↔ 1 or 2 or 3
2 ↔ 2 or 3
3 ↔ 3

Skew binary number system

Skew binary number system: wi = 2i+1 − 1 and Di = {0, 1, 2} for all i.

Why 2i+1 − 1? This is the size of perfect binary trees of height i + 1.

Why Di = {0, 1, 2}? Using {0, 1, 2} we cannot represent all numbers.

☞ This is a redundant number system, as well. However if we add the constraint
that only the lowest non-zero digit may be 2, then we regain uniqueness.

Salient properties: increment and decrement work in constant time.

Skew binary number system—continued

data Tree α = Leaf α | Node α (Tree α) (Tree α)

We use a so-called sparse representation to be able to access the lowest non-zero
digit in constant time.

type Size = Integer

data Seq α = Nil | Cons (Size,Tree α, Seq α)

cons :: ∀α . α → Seq α → Seq α
cons a (Cons (n1, t1,Cons (n2, t2, x )))
| n1 n2 = Cons (1 + n1 + n2,Node a t1 t2, x )

cons a x = Cons (1, Leaf a, x )

More number systems . . .

There is a lot more to discover:

I lazy number systems: good amortised bounds even in a persistent setting;

I segmented number systems: supports constant time deque operations;

I combinatorial number system: fixed number of digits;

I mixed-radix number systems: persistent arrays with sublogarithmic access;

I . . .

Exercises

Exercise 5. Explore the 1-2 fibonacci number system.

Exercise 6. Try to come up with a suitable definition of skew fibonacci numbers.

Outline of the talk

✔ Exploring the analogy (7–27)

✔ A toolbox of number systems (29–41)

✖ Analysis of data structures (43–51)

✖ A worked-out example: 2-3 finger trees (53–64)

Analysis of red-black trees

Problem: We insert an ascending sequence of elements into an empty red-black
tree. Which shape has the final tree?

Recap: red-black trees

Red-black trees were developed by R. Bayer under the name symmetric binary
B-trees as binary tree representations of 2-3-4 trees.

The idea of red-black trees is to represent 3- and 4-nodes by small binary trees,
which consist of a black root and one or two auxiliary red children.

a1

t1 t2

a2

a1

t1 t2

t3

a1

t1 a2

t2 t3

a2

a1

t1 t2

a3

t3 t4

Red condition: Each red node has a black parent.

Black condition: Each path from the root to an empty node contains exactly
the same number of black nodes (this is called the tree’s black height).

Red-black trees: balancing
a3

a2

a1

t1 t2

t3

t4

a1

t1 a3

a2

t2 t3

t4

↘ ↙
a2

a1

t1 t2

a3

t3 t4

↗ ↖
a3

a1

t1 a2

t2 t3

t4

a1

t1 a2

t2 a3

t3 t4

Examples

The following trees are generated for 1 6 i 6 8 (a new node is always coloured
red).

Insertion always traverses the left spine of the tree to the leftmost leaf.

Observations

☞ All the nodes below the left spine are black.

Re-consider the balancing operation—for emphasis we draw the left spine
horizontally.

t1 a1

t2

a2

t3

a3

t4

7−→

t1 a1

t2

a2

a3

t3 t4

☞ The black condition implies that the trees below the left spine must be
perfectly balanced. Thus, the generated red-black trees correspond to sequences
of pennants.

Observations—continued

It is helpful to redraw the examples according to the left-spine view.

Let h be the height of the rightmost pennant; the black condition implies that a
pennant of height i appears either once or twice for all 0 6 i 6 h.

☞ The generated red-black trees correspond to binary numbers in the 1-2
number system.

Construction of red-black trees

☞ The analogy to the 1-2 number system can be exploited to construct
red-black trees from ordered sequences of elements in linear time.

data Tree α = Empty
| Red (Tree α) α (Tree α)
| Black (Tree α) α (Tree α)

data Set α = Nil
| One (α,Tree α) (Set α)
| Two (α,Tree α) (α,Tree α) (Set α)

☞ We do not use a nested data type, so that we can easily transform the
left-spine view into an ordinary red-black tree.

Construction of red-black trees—continued

cons :: ∀α . (α,Tree α) → Set α → Set α
cons (a, t) Nil = One (a, t) Nil
cons (a1, t1) (One (a2, t2) x ) = Two (a1, t1) (a2, t2) x
cons (a1, t1) (Two (a2, t2) (a3, t3) x )

= One (a1, t1) (cons (a2,Black t2 a3 t3) x )

insert :: ∀α . α → Set α → Set α
insert a x = cons (a,Empty) x

Exercises

Exercise 4. Repeat the analysis with 1-2 brother trees.

Outline of the talk

✔ Exploring the analogy (7–27)

✔ A toolbox of number systems (29–41)

✔ Analysis of data structures (43–51)

✖ A worked-out example: 2-3 finger trees (53–64)

A more symmetric design

All the data structures we have seen so far are asymmetric:

I adding an element to the left end is easy and efficient;

I adding an element to the right end is difficult (if not impossible) and
inefficient.

☞ 2-3 finger trees remedy the situation by using two digits per level, one for the
left and one for the right end. (A finger provides efficient access to nodes of a
tree near a distinguished location.)

2-3 nodes

Finger trees are constructed from 2-3 nodes:

data Node α = Node2 α α | Node3 α α α

☞ 2-3 nodes contain two or three subtrees, but no keys; all data is stored in the
leaves of the tree.

2-3 trees

Conventional 2-3 trees can be defined as follows:

data Tree α = Zero α | Succ (Tree (Node α))

A typical 2-3 tree:

t h i s i s n o t a t r e e

All leaves of a 2-3 tree are at the same depth, the left and right spines have the
same length.

2-3 finger trees

If we take hold of the end nodes of a 2-3 tree and lift them up together, we obtain
a tree that looks like this:

t h

i s i s n o t a t

r e e

Each pair of nodes on the central spine is merged into a single node (called Deep).

☞ A sequence of n elements is represented by a tree of depth Θ(log n); an
element d positions from the nearest end is stored at a depth of Θ(log d).

2-3 finger trees—continued

Finger trees provide efficient access to both ends of a sequence:

data FingerTree α = Empty
| Single α
| Deep (Digit α) (FingerTree (Node α)) (Digit α)

The top level contains elements of type α, the next of type Node α, and so on:
the nth level contains elements of type Noden α, namely 2-3 trees of depth n.

A digit is a buffer of one to four elements, represented as a list to simplify the
presentation.

type Digit α = [α ]

Finger trees—deque operations

Adding a new element to the left of the sequence:

(/) :: α → FingerTree α → FingerTree α
a / Empty = Single a
a / Single b = Deep [a ] Empty [b ]
a / Deep [b, c, d , e ] m sf = Deep [a, b ] (Node3 c d e / m) sf
a / Deep pr m sf = Deep ([a ] ++ pr ) m sf

Finger trees—deque operations—continued

Adding to the right end of the sequence is the mirror image of ‘/’:

(.) :: FingerTree a → a → FingerTree a
Empty . a = Single a
Single b . a = Deep [b ] Empty [a ]
Deep pr m [e, d , c, b ] . a = Deep pr (m . Node3 e d c) [b, a ]
Deep pr m sf . a = Deep pr m (sf ++ [a ])

We classify digits of two or three elements (which correspond to nodes) as safe,
and those of one or four elements as dangerous (cf 0-1-2 random-access lists).

☞ A deque operation may only propagate to the next level from a dangerous
digit, but in doing so it makes that digit safe, so that the next operation to reach
that digit will not propagate.

Finger trees—concatenation

The use of 2-3 nodes rather than pairs provides enough flexibility so that we can
also append two 2-3 finger trees.

☞ The tree Empty will be an identity for concatenation, and Singles may be
concatenated using ‘/’ or ‘.’, so the only difficult case is concatenation of two
Deep trees.

Finger trees—concatenation—continued

Concatenation of two Deep trees:

Deep pr 1 m1 sf 1 ./ Deep pr 2 m2 sf 2 = Deep pr 1 . . . sf 2

We can use the prefix of the first tree as the prefix of the result, and the suffix of
the second tree as the suffix of the result.

To combine the rest to make the new middle subtree, we require a function of type

FingerTree (Node α) → Digit α → Digit α → FingerTree (Node α) →
FingerTree (Node α)

For simplicity, we combine the two digit arguments into a list of Nodes.

Finger trees—concatenation—continued

app3 :: FingerTree α → [α ] → FingerTree α → FingerTree α
app3 Empty ts xs = ts /′ xs
app3 xs ts Empty = xs .′ ts
app3 (Single x ) ts xs = x / (ts /′ xs)
app3 xs ts (Single x ) = (xs .′ ts) . x
app3 (Deep pr 1 m1 sf 1) ts (Deep pr 2 m2 sf 2)

= Deep pr 1 (app3 m1 (nodes (sf 1 ++ ts ++ pr 2)) m2) sf 2

☞ nodes groups a list of at most 12 elements into a list of 2-3 nodes, ‘/′’ adds
a list of at most 4 elements to the left of a finger tree.

(./) :: FingerTree α → FingerTree α → FingerTree α
xs ./ ys = app3 xs [ ] ys

☞ The recursion terminates when we reach the bottom of the shallower tree, so
the total time taken is Θ(log(min{n1, n2})).

Finger trees—concatenation—continued

nodes :: [α ] → [Node α ]
nodes [a, b ] = [Node2 a b ]
nodes [a, b, c ] = [Node3 a b c ]
nodes [a, b, c, d ] = [Node2 a b,Node2 c d ]
nodes (a : b : c : xs) = Node3 a b c : nodes xs

(/′) :: [a ] → FingerTree a → FingerTree a
x /′ t = foldr (/) t x

(.′) :: FingerTree a → [a ] → FingerTree a
t .′ x = foldl (.) t x

2-3 finger search trees

2-3 finger trees may also serve as implementations of

I priority queues,

I search trees,

I priority search queues,

I . . .

Idea: augment the internal nodes by additional information (size, split key, or
both) to steer the search.

☞ The use of 2-3 trees as opposed to pairs is essential: one can prove a lower
bound of Ω(

√
n) for insertion and deletion if the data structure is uniquely

determined by its size.

Outline of the talk

✔ Exploring the analogy (7–27)

✔ A toolbox of number systems (29–41)

✔ Analysis of data structures (43–51)

✔ A worked-out example: 2-3 finger trees (53–64)

Conclusion

I Number systems serve admirably as templates for data structures.

I Even the most exotic number systems seem to have their uses in the realm of
data structures.

I We have only touched the surface of this exciting topic.

Thanks for listening!

